5.D.2. Detailed Balance

(5.53) implies that $P(n,t)$ is time-independent, i.e., $\frac{\partial}{\partial t} P(n,t) = 0$, if

$$P(m,t)\, w_{mn}(t) = P(n,t)\, w_{nm}(t) \qquad \forall\ m \qquad (5.60a)$$

For processes with time-independent w_{mn} (and hence $\boldsymbol{W}(t) = \boldsymbol{W}$), the stationary state $\langle p^s \mid$ is given by [see (5.58)]

$$\langle p^s \mid \boldsymbol{W} = 0$$

i.e., it is a left eigenvector of the transition matrix $\boldsymbol{W}$.

Setting

$$P^s(n) = \langle p^s \mid n \rangle \equiv \lim_{t\to\infty} P(n,t)$$

turns (5.60a) into

$$P^s(m)\, w_{mn} = P^s(n)\, w_{nm} \qquad \forall\ m, n \qquad (5.60)$$

which is called detailed balance since it means during any Δt,

average number of transitions from m into n

= average number of transitions out of n back to m.

Note that given $\boldsymbol{W}$, $P^s(n)$ can be obtained by iteration using (5.60). For example, expressing all $P^s(n)$'s in terms of $P^s(1)$, we have

$$P^s(2) = P^s(1)\, \frac{w_{1,2}}{w_{2,1}}$$

$$P^s(3) = P^s(2)\, \frac{w_{2,3}}{w_{3,2}} = P^s(1)\, \frac{w_{1,2}}{w_{2,1}}\, \frac{w_{2,3}}{w_{3,2}}$$

and so on. The normalization

$$\sum_{n=1}^{M} P^s(n) = 1$$

then becomes a linear equation for $P^s(1)$ and be solved trivially.

(5.60) can be written in a symmetric form by dividing it with $\sqrt{P^s(m)\, P^s(n)}$ so that

$$\sqrt{\frac{P^s(m)}{P^s(n)}}\, w_{mn} = \sqrt{\frac{P^s(n)}{P^s(m)}}\, w_{nm} \qquad (5.60b)$$

Note that (5.60b) is just an identity, $w_{mm} = w_{mm}$, for $n = m$.

Let

$$V_{mn} = \sqrt{\frac{P^s(m)}{P^s(n)}}\, w_{mn} \qquad \forall\ m \neq n \qquad (5.61a)$$

Setting $m \leftrightarrow n$ gives

$$V_{nm} = \sqrt{\frac{P^s(n)}{P^s(m)}}\, w_{nm} \qquad \forall\ m \neq n$$

so that (5.61) becomes

$$V_{mn} = V_{nm} \qquad \forall\ m \neq n \qquad (5.61b)$$

From (5.55), we see that for $m \neq n$, $w_{mn} = W_{mn}$ so that by setting

$$V_{mn} = \sqrt{\frac{P^s(m)}{P^s(n)}}\, W_{mn} \qquad \forall\ m, n$$

$$= \sqrt{\frac{P^s(m)}{P^s(n)}} \left(w_{mn} - \delta_{mn} \sum_{k=1}^{M} w_{mk} \right) \tag{5.61}$$

(5.61b) is still valid. What's new is that we now have

$$V_{mm} = W_{mm} = w_{mm} - \sum_{k=1}^{M} w_{mk}$$

(5.56) then becomes

$$\frac{\partial P(n,t)}{\partial t} = \sum_{m=1}^{M} P(m,t) \sqrt{\frac{P^s(n)}{P^s(m)}}\, V_{mn}$$

$$\rightarrow \quad \frac{\partial \tilde{P}(n,t)}{\partial t} = \sum_{m=1}^{M} \tilde{P}(m,t)\, V_{mn} \tag{5.62}$$

where

$$\tilde{P}(n,t) = \frac{P(n,t)}{\sqrt{P^s(n)}}$$

In terms of the Dirac notations, (5.62) becomes

$$\frac{\partial \langle \tilde{p}(t) \mid n \rangle}{\partial t} = \sum_{m=1}^{M} \langle \tilde{p}(t) \mid m \rangle \langle m \mid \boldsymbol{V} \mid n \rangle$$

$$= \langle \tilde{p}(t) \mid \boldsymbol{V} \mid n \rangle \tag{5.62a}$$

where

$$\langle \tilde{p}(t) \mid n \rangle = \tilde{P}(n,t) \qquad\qquad \langle m \mid \boldsymbol{V} \mid n \rangle = V_{mn}$$

Since (5.62a) is true for all n, we have

$$\frac{\partial \langle \tilde{p}(t) \mid}{\partial t} = \langle \tilde{p}(t) \mid \boldsymbol{V} \tag{5.62b}$$

Since $\boldsymbol{V}$ is time-independent, the solution to (5.62b) is simply

$$\left\langle \tilde{p}(t) \right| = \left\langle \tilde{p}(0) \right| e^{\boldsymbol{V} t} \tag{5.63}$$

Since $\boldsymbol{V}$ is a real symmetric matrix, the left and right eigenvectors of the same non-degenerate eigenvalue are related by a transpose operation, i.e.,

$$\boldsymbol{V} \mid \psi_i \rangle = \lambda_i \mid \psi_i \rangle \qquad\qquad \langle \psi_i \mid \boldsymbol{V} = \langle \psi_i \mid \lambda_i$$

For degenerate eigenvalues, they can be chosen as transpose pairs.

Furthermore, it always has a complete orthonormal set of eigenvectors so that

$$\langle \psi_i \mid \psi_j \rangle = \delta_{ij} \qquad\qquad \sum_{i=0}^{M-1} \mid \psi_i \rangle \langle \psi_i \mid = 1$$

[See (5.67) for the reason of the choice of the range of i.]

Thus,

$$\boldsymbol{V} = \sum_{i=0}^{M-1} \lambda_i \mid \psi_i \rangle \langle \psi_i \mid$$

$$\rightarrow \quad e^{\boldsymbol{V} t} = \sum_{i=0}^{M-1} e^{\lambda_i t} \left| \psi_i \right\rangle \langle \psi_i \mid$$

and (5.63) becomes

$$\langle \tilde{p}(t) \mid = \sum_{i=0}^{M-1} \langle \tilde{p}(0) \mid \psi_i \rangle e^{\lambda_i t} \langle \psi_i \mid \tag{5.64}$$

$$\rightarrow \qquad \langle \tilde{p}(t) \mid n \rangle = \sum_{i=0}^{M-1} \sum_{m=1}^{M} \langle \tilde{p}(0) \mid m \rangle \langle m \mid \psi_i \rangle e^{\lambda_i t} \langle \psi_i \mid n \rangle$$

or
$$\tilde{P}(n, t) = \sum_{i=0}^{M-1} \sum_{m=1}^{M} \tilde{P}(m, 0)\, \psi_i(m)\, e^{\lambda_i t}\, \psi_i(n) \tag{5.65a}$$

where $\psi_i(m) = \langle m \mid \psi_i \rangle$ is the m^{th} component of the (column) eigenvector $\mid \psi_i \rangle$. In terms of the probability, we have

$$P(n, t) = \sum_{i=0}^{M-1} \sum_{m=1}^{M} \sqrt{\frac{P^s(n)}{P^s(m)}}\, P(m, 0)\, \psi_i(m)\, e^{\lambda_i t}\, \psi_i(n) \tag{5.65}$$

In order for $P(n, t)$ to remain finite as $t \to \infty$, we must have $\lambda_i \le 0$ for all i. In which case, property (5.57) guarantees an eigenvalue $\lambda_0 = 0$ so that (5.65) gives

$$P^s(n) = \sum_{m=1}^{M} \sqrt{\frac{P^s(n)}{P^s(m)}}\, P(m, 0)\, \psi_0(m)\, \psi_0(n) \tag{5.65b}$$

$$\rightarrow \qquad \frac{\sqrt{P^s(n)}}{\psi_0(n)} = \sum_{m=1}^{M} \frac{P(m, 0)}{\sqrt{P^s(m)}}\, \psi_0(m) = \text{const} \qquad \forall\ n \tag{5.65c}$$

Exercise 5.2.

Consider an asymmetric random walk on an open-ended lattice with 4 lattice sites. The transition rates are

$$w_{1,2} = w_{4,3} = 1 \qquad\qquad w_{2,3} = w_{3,4} = \frac{3}{4} \qquad\qquad w_{2,1} = w_{3,2} = \frac{1}{4}$$

and $w_{ij} = 0$ otherwise.

(a) Write the transition matrix $\boldsymbol{W}$ and show that it obeys detailed balance.

(b) Compute $\boldsymbol{V}$ and find its eigenvalues & eigenvectors.

(c) Write $P(n, t)$ for the case $P(n, 0) = \delta_{n1}$. What is $P(2, t)$?

Ans. (a)

$$\boldsymbol{W} = \begin{pmatrix} -1 & 1 & 0 & 0 \\ \frac{1}{4} & -1 & \frac{3}{4} & 0 \\ 0 & \frac{1}{4} & -1 & \frac{3}{4} \\ 0 & 0 & 1 & -1 \end{pmatrix}$$

From the *Mathematica* code below, we get, in order of decreasing magnitude,

$$\lambda_0 = 0 \qquad\qquad \lambda_1 = -\frac{1}{4}\left(4 - \sqrt{3}\right) \qquad\qquad \lambda_2 = -2\ \lambda_3 = -\frac{1}{4}\left(4 + \sqrt{3}\right)$$

The corresponding (unnormalized) right eigenvectors are

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \end{pmatrix} \qquad \begin{pmatrix} -3\sqrt{3} \\ -\frac{9}{4} \\ \frac{\sqrt{3}}{4} \\ 1 \end{pmatrix} \qquad \begin{pmatrix} -1 \\ 1 \\ -1 \\ 1 \end{pmatrix} \qquad \begin{pmatrix} 3\sqrt{3} \\ -\frac{9}{4} \\ -\frac{\sqrt{3}}{4} \\ 1 \end{pmatrix}$$

The corresponding (unnormalized) left eigenvectors are

$$\begin{pmatrix} \frac{1}{9} & \frac{4}{9} & \frac{4}{3} & 1 \end{pmatrix}, \quad \begin{pmatrix} -\frac{1}{\sqrt{3}} & -1 & \frac{1}{\sqrt{3}} & 1 \end{pmatrix}, \quad \begin{pmatrix} -\frac{1}{9} & \frac{4}{9} & -\frac{4}{3} & 1 \end{pmatrix}, \quad \begin{pmatrix} \frac{1}{\sqrt{3}} & -1 & -\frac{1}{\sqrt{3}} & 1 \end{pmatrix}$$

The stationary state $\langle p^s \mid$ is just the left eigenvector for λ_0 normalized to $\sum_{n=1}^{M} \langle p^s \mid n \rangle = 1$.

$$\langle p^s \mid = \langle \chi_0 \mid \; = \frac{9}{26} \begin{pmatrix} \frac{1}{9} & \frac{4}{9} & \frac{4}{3} & 1 \end{pmatrix} = \begin{pmatrix} \frac{1}{26} & \frac{2}{13} & \frac{6}{13} & \frac{9}{26} \end{pmatrix}$$

$$P^s(n) = \langle p^s \mid n \rangle = \langle \chi_0 \mid n \rangle$$

Interestingly, $\sum_{n=1}^{M} \langle \chi_j \mid n \rangle = 0$ for $j = 1, 2, 3$.

The detailed balance

$$P^s(n)\, w_{n m} = P^s(m)\, w_{m n}$$

is equivalent to saying that the matrix $\boldsymbol{A}$ with elements

$$A_{n m} = P^s(n)\, w_{n m} \qquad \to \qquad A_{m n} = P^s(m)\, w_{m n}$$

is symmetric.

As shown in §Code (a),

$$\boldsymbol{A} = \begin{pmatrix} -\frac{1}{26} & \frac{1}{26} & 0 & 0 \\ \frac{1}{26} & -\frac{2}{13} & \frac{3}{26} & 0 \\ 0 & \frac{3}{26} & -\frac{6}{13} & \frac{9}{26} \\ 0 & 0 & \frac{9}{26} & -\frac{9}{26} \end{pmatrix}$$

is indeed symmetric.

Code (a)

```
W = {{-1, 1, 0, 0}, {1/4, -1, 3/4, 0}, {0, 1/4, -1, 3/4}, {0, 0, 1, -1}};

(* eigenvalues  & right eigenvectors |ψi⟩ *)
{λ, rv} = Eigensystem[W]
```

$$\left\{\left\{-2, \frac{1}{4}\left(-4-\sqrt{3}\right), \frac{1}{4}\left(-4+\sqrt{3}\right), 0\right\},\right.$$
$$\left.\left\{\{-1, 1, -1, 1\}, \left\{3\sqrt{3}, -\frac{9}{4}, -\frac{\sqrt{3}}{4}, 1\right\}, \left\{-3\sqrt{3}, -\frac{9}{4}, \frac{\sqrt{3}}{4}, 1\right\}, \{1, 1, 1, 1\}\right\}\right\}$$

```
(* rows of rv are right eigenvectors *)
Table[W.rv[[i]] == λ[[i]] rv[[i]], {i, 4}]

{True, True, True, True}
```

```
(* columns of rvᵀ are right eigenvectors *)
rvᵀ // MatrixForm
```

$$\begin{pmatrix} -1 & 3\sqrt{3} & -3\sqrt{3} & 1 \\ 1 & -\frac{9}{4} & -\frac{9}{4} & 1 \\ -1 & -\frac{\sqrt{3}}{4} & \frac{\sqrt{3}}{4} & 1 \\ 1 & 1 & 1 & 1 \end{pmatrix}$$

```
(* eigenvalues  & left eigenvectors ⟨χi| *)
{λ, lv} = Eigensystem[Wᵀ]
```

$$\left\{\left\{-2, \frac{1}{4}\left(-4-\sqrt{3}\right), \frac{1}{4}\left(-4+\sqrt{3}\right), 0\right\},\right.$$
$$\left.\left\{\left\{-\frac{1}{9}, \frac{4}{9}, -\frac{4}{3}, 1\right\}, \left\{\frac{1}{\sqrt{3}}, -1, -\frac{1}{\sqrt{3}}, 1\right\}, \left\{-\frac{1}{\sqrt{3}}, -1, \frac{1}{\sqrt{3}}, 1\right\}, \left\{\frac{1}{9}, \frac{4}{9}, \frac{4}{3}, 1\right\}\right\}\right\}$$

```
(* Stationary state ⟨ pˢ| *)
lvnC = Plus @@ lv[[4]]
ps = lv[[4]] / lvnC
```

$$\frac{26}{9}$$

$$\left\{\frac{1}{26}, \frac{2}{13}, \frac{6}{13}, \frac{9}{26}\right\}$$

```
(* Anm = Pˢ(n) wnm *)
(A = Table[ ps[[n]] W[[n, m]], {n, 4}, {m, 4} ]) // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{26} & \frac{1}{26} & 0 & 0 \\ \frac{1}{26} & -\frac{2}{13} & \frac{3}{26} & 0 \\ 0 & \frac{3}{26} & -\frac{6}{13} & \frac{9}{26} \\ 0 & 0 & \frac{9}{26} & -\frac{9}{26} \end{pmatrix}$$

```
(* rows of lv are left eigenvectors *)
Table[lv[[i]].W == λ[[i]] lv[[i]], {i, 4}]
```

{True, True, True, True}

```
lv // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{9} & \frac{4}{9} & -\frac{4}{3} & 1 \\ \frac{1}{\sqrt{3}} & -1 & -\frac{1}{\sqrt{3}} & 1 \\ -\frac{1}{\sqrt{3}} & -1 & \frac{1}{\sqrt{3}} & 1 \\ \frac{1}{9} & \frac{4}{9} & \frac{4}{3} & 1 \end{pmatrix}$$

```
(* orthogonality ⟨ χi|ψj ⟩ ∝ δij  *)
(* left eigenvector in row form, right eigenvector in column form *)
(nC = lv.rvᵀ) // MatrixForm
```

$$\begin{pmatrix} \frac{26}{9} & 0 & 0 & 0 \\ 0 & \frac{13}{2} & 0 & 0 \\ 0 & 0 & \frac{13}{2} & 0 \\ 0 & 0 & 0 & \frac{26}{9} \end{pmatrix}$$

```
(* normalized left eigenvectors ⟨χi| → ⟨χi| / ⟨χi|ψi⟩ *)
lvn = Table[lv[[i]] / nC[[i, i]], {i, 4}]
```

$$\left\{\left\{-\frac{1}{26}, \frac{2}{13}, -\frac{6}{13}, \frac{9}{26}\right\}, \left\{\frac{2}{13\sqrt{3}}, -\frac{2}{13}, -\frac{2}{13\sqrt{3}}, \frac{2}{13}\right\}, \left\{-\frac{2}{13\sqrt{3}}, -\frac{2}{13}, \frac{2}{13\sqrt{3}}, \frac{2}{13}\right\}, \left\{\frac{1}{26}, \frac{2}{13}, \frac{6}{13}, \frac{9}{26}\right\}\right\}$$

```
(* orthonormality ⟨ χi|ψj ⟩ = δij *)
(* left eigenvector in row form, right eigenvector in column form *)
(nC = lvn. rvᵀ) // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(* completeness ∑i |ψi⟩⟨χi|= I *)
(* right eigenvector in row form, left eigenvector in column form *)
rv.lvnᵀ // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

Ans. (b)

From §Code (b), we have

$$\boldsymbol{V} = \begin{pmatrix} -1 & \frac{1}{2} & 0 & 0 \\ \frac{1}{2} & -1 & \frac{\sqrt{3}}{4} & 0 \\ 0 & \frac{\sqrt{3}}{4} & -1 & \frac{\sqrt{3}}{2} \\ 0 & 0 & \frac{\sqrt{3}}{2} & -1 \end{pmatrix}$$

Also, in order of decreasing magnitude,

$$\lambda_0 = 0 \qquad \lambda_1 = -\frac{1}{4}\left(4-\sqrt{3}\right) \qquad \lambda_2 = -2\ \lambda_3 = -\frac{1}{4}\left(4+\sqrt{3}\right)$$

The corresponding (unnormalized) eigenvectors are

$$\begin{pmatrix} \frac{1}{3} \\ \frac{2}{3} \\ \frac{2}{\sqrt{3}} \\ 1 \end{pmatrix} \quad \begin{pmatrix} -\sqrt{3} \\ -\frac{3}{2} \\ \frac{1}{2} \\ 1 \end{pmatrix} \quad \begin{pmatrix} -\frac{1}{3} \\ \frac{2}{3} \\ -\frac{2}{\sqrt{3}} \\ 1 \end{pmatrix} \quad \begin{pmatrix} \sqrt{3} \\ -\frac{3}{2} \\ -\frac{1}{2} \\ 1 \end{pmatrix}$$

With the normalization

$$|\,\psi_i\rangle \to \frac{|\,\psi_i\rangle}{\sqrt{\langle\,\psi_i\,|\,\psi_i\,\rangle}}$$

the normalized eigenvectors are

$$|\,\psi_0\rangle = \begin{pmatrix} \frac{1}{\sqrt{26}} \\ \sqrt{\frac{2}{13}} \\ \sqrt{\frac{6}{13}} \\ \frac{3}{\sqrt{26}} \end{pmatrix} \quad |\,\psi_1\rangle = \begin{pmatrix} -\sqrt{\frac{6}{13}} \\ -\frac{3}{\sqrt{26}} \\ \frac{1}{\sqrt{26}} \\ \sqrt{\frac{2}{13}} \end{pmatrix} \quad |\,\psi_2\rangle = \begin{pmatrix} -\frac{1}{\sqrt{26}} \\ \sqrt{\frac{2}{13}} \\ -\sqrt{\frac{6}{13}} \\ \frac{3}{\sqrt{26}} \end{pmatrix} \quad |\,\psi_3\rangle = \begin{pmatrix} \sqrt{\frac{6}{13}} \\ -\frac{3}{\sqrt{26}} \\ -\frac{1}{\sqrt{26}} \\ \sqrt{\frac{2}{13}} \end{pmatrix}$$

Code (b)

```
V = Table[Sqrt[ps[[n]]/ps[[m]]] W[[n, m]], {n, 4}, {m, 4}];
V // MatrixForm
```

$$\begin{pmatrix} -1 & \frac{1}{2} & 0 & 0 \\ \frac{1}{2} & -1 & \frac{\sqrt{3}}{4} & 0 \\ 0 & \frac{\sqrt{3}}{4} & -1 & \frac{\sqrt{3}}{2} \\ 0 & 0 & \frac{\sqrt{3}}{2} & -1 \end{pmatrix}$$

```
(* eigenvalues & eigenvectors |ψi⟩ *)
{λp, ev} = Eigensystem[V]
```

$$\left\{\left\{-2, \frac{1}{4}\left(-4-\sqrt{3}\right), \frac{1}{4}\left(-4+\sqrt{3}\right), 0\right\},\right.$$
$$\left.\left\{\left\{-\frac{1}{3}, \frac{2}{3}, -\frac{2}{\sqrt{3}}, 1\right\}, \left\{\sqrt{3}, -\frac{3}{2}, -\frac{1}{2}, 1\right\}, \left\{-\sqrt{3}, -\frac{3}{2}, \frac{1}{2}, 1\right\}, \left\{\frac{1}{3}, \frac{2}{3}, \frac{2}{\sqrt{3}}, 1\right\}\right\}\right\}$$

```
(* orthogonality ⟨ ψi|ψj ⟩ ∝ δij  *)
(nCp = ev.ev^T) // MatrixForm
```

$$\begin{pmatrix} \frac{26}{9} & 0 & 0 & 0 \\ 0 & \frac{13}{2} & 0 & 0 \\ 0 & 0 & \frac{13}{2} & 0 \\ 0 & 0 & 0 & \frac{26}{9} \end{pmatrix}$$

```
(* normalized  eigenvectors |ψi 〉 → |ψi 〉/Sqrt[〈 ψi| ψi 〉]  *)
evn = Table[ev[[i]] / Sqrt[nCp[[i, i]]], {i, 4}]
```

$$\left\{\left\{-\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}, -\sqrt{\frac{6}{13}}, \frac{3}{\sqrt{26}}\right\}, \left\{\sqrt{\frac{6}{13}}, -\frac{3}{\sqrt{26}}, -\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}\right\}, \left\{-\sqrt{\frac{6}{13}}, -\frac{3}{\sqrt{26}}, \frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}\right\}, \left\{\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}, \sqrt{\frac{6}{13}}, \frac{3}{\sqrt{26}}\right\}\right\}$$

Ans. (c)

For the initial condition $P(n, 0) = \delta_{n\,1}$, (5.65) becomes

$$P(n, t) = P^s(n) + \sum_{i=1}^{3} \sqrt{\frac{P^s(n)}{P^s(1)}}\ \psi_i(1)\ e^{\lambda_i t}\ \psi_i^*(n)$$

Hence,

$$P(2, t) = P^s(2) + \sum_{i=1}^{3} \sqrt{\frac{P^s(2)}{P^s(1)}}\ \psi_i(1)\ e^{\lambda_i t}\ \psi_i^*(2)$$

$$= \frac{2}{13} - \frac{2\,e^{-2\,t}}{13} - \frac{6}{13}\sqrt{3}\ e^{\frac{1}{4}\left(-4-\sqrt{3}\right)t} + \frac{6}{13}\sqrt{3}\ e^{\frac{1}{4}\left(-4+\sqrt{3}\right)t}$$

$$= 0.153846 - 0.153846\,e^{-2.\,t} - 0.799408\,e^{-1.43301\,t} + 0.799408\,e^{-0.566987\,t}$$

Code (c)

```
ps
```

$$\left\{\frac{1}{26}, \frac{2}{13}, \frac{6}{13}, \frac{9}{26}\right\}$$

```
λp
```

$$\left\{-2, \frac{1}{4}\left(-4-\sqrt{3}\right), \frac{1}{4}\left(-4+\sqrt{3}\right), 0\right\}$$

```
evn
```

$$\left\{\left\{-\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}, -\sqrt{\frac{6}{13}}, \frac{3}{\sqrt{26}}\right\}, \left\{\sqrt{\frac{6}{13}}, -\frac{3}{\sqrt{26}}, -\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}\right\}, \left\{-\sqrt{\frac{6}{13}}, -\frac{3}{\sqrt{26}}, \frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}\right\}, \left\{\frac{1}{\sqrt{26}}, \sqrt{\frac{2}{13}}, \sqrt{\frac{6}{13}}, \frac{3}{\sqrt{26}}\right\}\right\}$$

$$\text{ps}[[2]] + \sum_{i=1}^{3} \sqrt{\frac{\text{ps}[[2]]}{\text{ps}[[1]]}}\ \text{evn}[[i, 1]]\ e^{\lambda p[[i]]\,t}\ \text{evn}[[i, 2]]$$

$$\frac{2}{13} - \frac{2\,e^{-2\,t}}{13} - \frac{6}{13}\sqrt{3}\ e^{\frac{1}{4}\left(-4-\sqrt{3}\right)t} + \frac{6}{13}\sqrt{3}\ e^{\frac{1}{4}\left(-4+\sqrt{3}\right)t}$$

```
% // N
```

$0.153846 - 0.153846\, e^{-2.\, t} - 0.799408\, e^{-1.43301\, t} + 0.799408\, e^{-0.566987\, t}$

Exercise 5.3.

Consider an asymmetric random walk on a periodic lattice with 4 lattice sites. The transition rates are

$$w_{12} = w_{23} = w_{34} = w_{41} = \frac{3}{4} \qquad\qquad w_{21} = w_{32} = w_{43} = w_{14} = \frac{1}{4}$$

and $w_{ij} = 0$ otherwise.

Write the transition matrix $\boldsymbol{W}$ and show that this system does not obey detailed balance.

Answer

$$\boldsymbol{W} = \begin{pmatrix} -1 & \frac{3}{4} & 0 & \frac{1}{4} \\ \frac{1}{4} & -1 & \frac{3}{4} & 0 \\ 0 & \frac{1}{4} & -1 & \frac{3}{4} \\ \frac{3}{4} & 0 & \frac{1}{4} & -1 \end{pmatrix}$$

The eigenvalues of $\boldsymbol{W}$ are

$$\lambda_0 = 0 \qquad \lambda_1 = -1 + \frac{1}{2} i \qquad \lambda_2 = -1 - \frac{1}{2} i \qquad \lambda_3 = -2$$

with corresponding (unnormalized) right eigenvectors

$$\begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \end{pmatrix} \qquad \begin{pmatrix} i \\ -1 \\ -i \\ 1 \end{pmatrix} \qquad \begin{pmatrix} -i \\ -1 \\ i \\ 1 \end{pmatrix} \qquad \begin{pmatrix} -1 \\ 1 \\ -1 \\ 1 \end{pmatrix}$$

and left eigenvectors

$$(1 \;\; 1 \;\; 1 \;\; 1), \quad (-i \;\; -1 \;\; i \;\; 1), \quad (i \;\; -1 \;\; -i \;\; 1), \quad (-1 \;\; 1 \;\; -1 \;\; 1)$$

Since $\text{Re}\, \lambda_i \leq 0$ for all i, there is a long time stationary state given by the left eigenvector of $\lambda_0 = 0$.

Using the sum rule for probabilities, $\sum_{n=1}^{M} \langle p^s \mid n \rangle = 1$, we have

$$\langle p^s \mid = \langle \chi_0 \mid \; = \frac{1}{4} (1 \;\; 1 \;\; 1 \;\; 1) = \left(\tfrac{1}{4} \;\; \tfrac{1}{4} \;\; \tfrac{1}{4} \;\; \tfrac{1}{4} \right)$$

$$\rightarrow \qquad P^s(n) = \langle p^s \mid n \rangle = \langle \chi_0 \mid n \rangle \; = \frac{1}{4} \qquad\qquad \forall\, n$$

On the other hand, the normalization

$$\langle \chi_0 \mid \psi_0 \rangle = 1 \qquad \rightarrow \qquad \mid \psi_0 \rangle = \begin{pmatrix} 1 \\ 1 \\ 1 \\ 1 \end{pmatrix}$$

Therefore, (e) in §5.D.1 also gives

$$P^s(n) = \frac{1}{4} \sum_{m=1}^{M} P(m, 0) \; = \frac{1}{4} \qquad\qquad \forall\, n$$

The detailed balance

$$P^s(n)\, w_{n\,m} = P^s(m)\, w_{m\,n}$$

is equivalent to saying that the matrix $\boldsymbol{A}$ with elements

$$A_{n\,m} = P^s(n)\, w_{n\,m} \qquad \to \qquad A_{m\,n} = P^s(m)\, w_{m\,n}$$

is symmetric.

As shown in §Code,

$$\boldsymbol{A} = \begin{pmatrix} -\frac{1}{4} & \frac{3}{16} & 0 & \frac{1}{16} \\ \frac{1}{16} & -\frac{1}{4} & \frac{3}{16} & 0 \\ 0 & \frac{1}{16} & -\frac{1}{4} & \frac{3}{16} \\ \frac{3}{16} & 0 & \frac{1}{16} & -\frac{1}{4} \end{pmatrix}$$

is not symmetric so that detailed balance is not observed. Therefore,

$$\frac{\partial\, P(n,\, t)}{\partial\, t} \neq 0 \qquad \forall\; t$$

and we have a probability current around the lattice permanently.

Code

$$W = \begin{pmatrix} -1 & \frac{3}{4} & 0 & \frac{1}{4} \\ \frac{1}{4} & -1 & \frac{3}{4} & 0 \\ 0 & \frac{1}{4} & -1 & \frac{3}{4} \\ \frac{3}{4} & 0 & \frac{1}{4} & -1 \end{pmatrix}$$

$$\left\{\left\{-1, \frac{3}{4}, 0, \frac{1}{4}\right\}, \left\{\frac{1}{4}, -1, \frac{3}{4}, 0\right\}, \left\{0, \frac{1}{4}, -1, \frac{3}{4}\right\}, \left\{\frac{3}{4}, 0, \frac{1}{4}, -1\right\}\right\}$$

```
(* eigenvalues λi = rλ[[i]] & right eigenvectors |ψi⟩= rv[[i]] *)
{rλ, rv} = Eigensystem[W]
```

$$\left\{\left\{-2, -1+\frac{i}{2}, -1-\frac{i}{2}, 0\right\}, \{\{-1, 1, -1, 1\}, \{i, -1, -i, 1\}, \{-i, -1, i, 1\}, \{1, 1, 1, 1\}\}\right\}$$

```
(* eigenvalues λi = lλ[[i]] & left eigenvectors ⟨χi|= lv[[i]] *)
{lλ, lv} = Eigensystem[Wᵀ]
```

$$\left\{\left\{-2, -1+\frac{i}{2}, -1-\frac{i}{2}, 0\right\}, \{\{-1, 1, -1, 1\}, \{-i, -1, i, 1\}, \{i, -1, -i, 1\}, \{1, 1, 1, 1\}\}\right\}$$

```
(* Checking *)
Table[ lv[[i]].W == lλ[[i]] lv[[i]], {i, 4}]
```

{True, True, True, True}

```
(* orthogonality *)
Table[ lv[[i]].rv[[j]], {i, 4}, {j, 4}] // MatrixForm
```

$$\begin{pmatrix} 4 & 0 & 0 & 0 \\ 0 & 4 & 0 & 0 \\ 0 & 0 & 4 & 0 \\ 0 & 0 & 0 & 4 \end{pmatrix}$$

```
(* normalized left eigenvectors *)
lvn = Table[lv[[i]] / (lv[[i]].rv[[i]]), {i, 4}]
```

$$\left\{\left\{-\frac{1}{4},\frac{1}{4},-\frac{1}{4},\frac{1}{4}\right\},\left\{-\frac{i}{4},-\frac{1}{4},\frac{i}{4},\frac{1}{4}\right\},\left\{\frac{i}{4},-\frac{1}{4},-\frac{i}{4},\frac{1}{4}\right\},\left\{\frac{1}{4},\frac{1}{4},\frac{1}{4},\frac{1}{4}\right\}\right\}$$

```
(* orthonormality *)
Table[ lvn[[i]].rv[[j]], {i, 4}, {j, 4}] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(* |ψi⟩⟨ χi| *)
rlProd = Table[Outer[Times, rv[[i]], lvn[[i]]], {i, 4}];
MatrixForm /@ rlProd
```

$$\left\{\begin{pmatrix} \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{1}{4} & -\frac{1}{4} & \frac{1}{4} \end{pmatrix}, \begin{pmatrix} \frac{1}{4} & -\frac{i}{4} & -\frac{1}{4} & \frac{i}{4} \\ \frac{i}{4} & \frac{1}{4} & -\frac{i}{4} & -\frac{1}{4} \\ -\frac{1}{4} & \frac{i}{4} & \frac{1}{4} & -\frac{i}{4} \\ -\frac{i}{4} & -\frac{1}{4} & \frac{i}{4} & \frac{1}{4} \end{pmatrix}, \begin{pmatrix} \frac{1}{4} & \frac{i}{4} & -\frac{1}{4} & -\frac{i}{4} \\ -\frac{i}{4} & \frac{1}{4} & \frac{i}{4} & -\frac{1}{4} \\ -\frac{1}{4} & -\frac{i}{4} & \frac{1}{4} & \frac{i}{4} \\ \frac{i}{4} & -\frac{1}{4} & -\frac{i}{4} & \frac{1}{4} \end{pmatrix}, \begin{pmatrix} \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \\ \frac{1}{4} & \frac{1}{4} & \frac{1}{4} & \frac{1}{4} \end{pmatrix}\right\}$$

```
(* Completeness: ∑_{i=1}^3 |ψi⟩⟨ χi| = I *)
```

$$\sum_{i=1}^{4} \texttt{rlProd[[i]] // MatrixForm}$$

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(* Stationary state ⟨ p^s| *)
ps = lvn[[4]]
```

$$\left\{\frac{1}{4},\frac{1}{4},\frac{1}{4},\frac{1}{4}\right\}$$

```
(* A_nm = P^s(n) w_nm *)
(A = Table[ ps[[n]] W[[n, m]], {n, 4}, {m, 4} ]) // MatrixForm
```

$$\begin{pmatrix} -\frac{1}{4} & \frac{3}{16} & 0 & \frac{1}{16} \\ \frac{1}{16} & -\frac{1}{4} & \frac{3}{16} & 0 \\ 0 & \frac{1}{16} & -\frac{1}{4} & \frac{3}{16} \\ \frac{3}{16} & 0 & \frac{1}{16} & -\frac{1}{4} \end{pmatrix}$$